Solving differential equations

Differentiation

`D[f, x]`	partial derivative	$\frac{\partial}{\partial x} f$
`D[f, x1, x2, ... ]`	multiple derivative	$\frac{\partial}{\partial x_1} \frac{\partial}{\partial x_2} \ldots f$
`D[f, {x, n}]`	repeated derivative	$\frac{\partial^n f}{\partial x^n}$
`Dt[f]`	total differential	$d f$
`Dt[f, x]`	total derivative	$\frac{d}{d x} f$

`In[1]:= D[ ArcTan[x], x ]`

Differentiate with respect to x

`Out[1]=` $\frac{1}{1+x^2}$

`In[2]:= D[ x^n, {x, 3} ]`

Differentiate 3 times with respect to x

`Out[2]=` $(-2+n)(-1+n)\, n\, x^{-3+n}$

Integration

`Integrate[f, x]`	the indefinite integral	$\int f\, dx$
`Integrate[f, x, y]`	the multiple integral	$\int dx\, dy\, f$
`Integrate[f, {x, xmin, xmax}]`	the definite integral	$\int_{x_{min}}^{x_{max}} f\, dx$
`Integrate[f, {x, xmin, xmax}, {y, ymin, ymax}]`	the multiple integral	$\int_{x_{min}}^{x_{max}} dx \int_{y_{min}}^{y_{max}} dy\, f$

`In[1]:= Integrate[1/(x^4 - a^4), x]`

Integrate over x

`Out[1]=` $-\frac{\mathrm{ArcTan}\left[\frac{x}{a}\right]}{2 a^3} + \frac{\mathrm{Log}[a-x]}{4 a^3} - \frac{\mathrm{Log}[a+x]}{4 a^3}$

`In[2]:= Integrate[Exp[-x^2], {x, 0, Infinity}]`

Integrate over a defiend range

`Out[2]=` $\frac{\sqrt{\pi}}{2}$

Numerical Integration

`In[1]:= NIntegrate[Exp[-x^2], {x, -Infinity, Infinity}]`

Integrate over infinite regions

`Out[1]= 1.77245`

```
In[1]:= NIntegrate[ Sin[x y], {x, 0, 1}, {y, 0, x} ]
```

double Integral over a triangle

```
Out[1]=  0.119906
```

Solving Equations

Note:

```
x = y                  x to have value y
x == y                 tests whether x and y are equal
```

```
Solve[lhs == rhs, x]   solve an equation, giving a list of rules for x
x /. solution          use the list of rules to get values for x
expr /. solution       use the list of rules to get values for an expression
```

```
In[1]:= Solve[x^2 + 2x - 7 == 0, x]
```

solve an Equation

Out[1]= $\{\{x \to -1 - 2\sqrt{2}\}, \{x \to -1 + 2\sqrt{2}\}\}$

```
In[2]:= N[ % ]
```

get the numeric values

```
Out[2]=  {{x → -3.82843}, {x → 1.82843}}
```

You can get a list of the actual solutions for x by applying the rules generated by Solve to x using the replacement operator.

```
In[3]:= x /. %
Out[3]=  {-3.82843, 1.82843}
```

```
In[1]:= FindRoot[ Cos[x] ==x, {x, 0} ]
```

find an approximate numerical solution giving the starting value

```
Out[1]=  {x → 0.739085}
```

Numerical approximation in Equations

```
In[1]:= NSolve[ x^5 + x + 1 ==0, x ]
```

solve an equation numerically

```
         {{x → -0.754878}, {x → -0.5 - 0.866025 ⅈ},
          {x → -0.5 + 0.866025 ⅈ}, {x → 0.877439 - 0.744862 ⅈ},
Out[1]=   {x → 0.877439 + 0.744862 ⅈ}}
```

NSolve returns a list of rules. In order to get to “numbers”, one has to replace x by the solution list. We obtain a solution by applying the /. operator

Differential equations

```
DSolve[eqns, y[x], x]
solve a differential equation for y[x], taking x as the independentvariable
DSolve[eqns, y, x]
                                   give a solution for y in pure function form
```

Solution to the differential equation $y'(x) = a\,y(x) + 1$. C[1] is a coefficient which must be determined from boundary conditions.

```
In[1]:= DSolve[ y'[x] == a y[x] + 1, y[x], x ]
```

Out[1]= $\left\{\left\{y[x] \to -\frac{1}{a} + e^{a x} C[1]\right\}\right\}$

```
In[2]:= DSolve[ {y'[x] == a y[x] + 1, y[0] == 0}, y[x], x ]
```

Out[2]= $\left\{\left\{y[x] \to \frac{-1 + e^{a x}}{a}\right\}\right\}$

Always give differential equations explicitly in terms of functions such as y[x]

Numerical differential equations

Numerical approximations to functions are represented as InterpolatingFunctions objects.

```
In[1]:= NDSolve[{y'[x] ==y[x], y[0] ==1}, y, {x, 0, 2}]
```

solve a differential equation numerically

```
Out[1]= {{y → InterpolatingFunction[{{0., 2.}}, <>]}}
```

```
In[2]:=  y[1.5] /. %
```

obtain the value at a specific point

```
Out[2]= {4.48169}
```

```
In[3]:=  Plot[Evaluate[y[x] /. %1], {x, 0, 1}]
```

Plot the solution from line 1 in the range [0,1]